


SEND METHOD



This chapter covers the following topics:

- Function
- Syntax Description
- Example

For an explanation of the symbols used in the syntax diagram, see *Syntax Symbols*.

Related Statements: CREATE OBJECT | DEFINE CLASS | INTERFACE | METHOD | PROPERTY

Belongs to Function Group: *Component Based Programming*

Function

The SEND METHOD statement is used to invoke a particular method of an object. See the section *NaturalX* in the *Programming Guide* for information on component-based programming.

Syntax Description


Operand Definition Table:

Syntax Element	Description
<i>operand3</i>	<p>Parameter(s) Specific to the Method:</p> <p>As <i>operand3</i> you can specify parameters specific to the method.</p> <p>In the following example, the object #O3 has the method <code>PositionTo</code> with the parameter <code>Pos</code>. The method is called in the following way:</p> <pre>SEND 'PositionTo' TO #O3 WITH Pos</pre> <p>Methods can have optional parameters. Optional parameters need not to be specified when the method is called. To omit an optional parameter, use the placeholder <code>1X</code>. To omit <i>n</i> optional parameters, use the placeholder <code>nX</code>.</p> <p>In the following example, the method <code>SetAddress</code> of the object #O4 has the parameters <code>FirstName</code>, <code>MiddleInitial</code>, <code>LastName</code>, <code>Street</code> and <code>City</code>, where <code>MiddleInitial</code>, <code>Street</code> and <code>City</code> are optional. The following statements apply:</p> <pre>* Specifying all parameters. SEND 'SetAddress' TO #O4 WITH FirstName MiddleInitial LastName Street City * Omitting one optional parameter. SEND 'SetAddress' TO #O4 WITH FirstName 1X LastName Street City * Omitting all optional parameters. SEND 'SetAddress' TO #O4 WITH FirstName 1X LastName 2X</pre> <p>Omitting a non-optional (mandatory) parameter results in a runtime error.</p>
AD=	<p>Attribute Definition:</p> <p>If <i>operand3</i> is a variable, you can mark it in one of the following ways:</p> <pre>AD=O Non-modifiable, see session parameter AD=O. AD=M Modifiable, see session parameter AD=M. This is the default setting. AD=A Input only, see session parameter AD=A.</pre> <p>If <i>operand3</i> is a constant, AD cannot be explicitly specified. For constants AD=O always applies.</p>
<i>nX</i>	<p>Parameter(s) to be Skipped:</p> <p>With the notation <i>nX</i> you can specify that the next <i>n</i> parameters are to be skipped (for example, <code>1X</code> to skip the next parameter, or <code>3X</code> to skip the next three parameters). This means that for the next <i>n</i> parameters no values are passed to the method.</p> <p>For a method implemented in Natural, a parameter that is to be skipped must be defined with the keyword <code>OPTIONAL</code> in the dialog's <code>DEFINE DATA PARAMETER</code> statement. <code>OPTIONAL</code> means that a value can - but need not - be passed from the invoking object to such a parameter.</p>

Syntax Element	Description
RETURN <i>operand4</i>	<p>RETURN Clause:</p> <p>If the RETURN clause is omitted and the method has a return value, the return value is discarded.</p> <p>If the RETURN clause is specified, <i>operand4</i> contains the return value of the method. If the method execution fails, <i>operand4</i> is reset to its initial value.</p> <p>Note: For classes written in Natural, the return value of a method is defined by entering one additional parameter in the parameter data area of the method and by marking it with BY VALUE RESULT. For more information, see the PARAMETER clause in the INTERFACE statement description. Therefore the parameter data area of a method that is written in Natural and that has a return value always contains one additional field next to the method parameters. This is to be considered when you call a method of a Natural written class and want to use the parameter data area of the method in the SEND statement.</p>
GIVING <i>operand5</i>	<p>GIVING Clause:</p> <p>If the GIVING clause is not specified, the Natural run time error processing is triggered if an error occurs.</p> <p>If the GIVING clause is specified, <i>operand5</i> contains the Natural message number if an error occurred, or zero on success.</p>

Example

The following diagram gives an overview of the programming objects that are used in this example. The corresponding source code and the program output are shown below. Links to the source code of an object are provided in the diagram.



Program METH01 - CREATE OBJECT and SEND METHOD Using a Class and Several Methods:

```

** Example 'METH01':  CREATE OBJECT and SEND METHOD
**                   using a class and several methods (see METH*)
*****
DEFINE DATA
LOCAL
  USING METHA
LOCAL
1 L-STUDENT HANDLE OF OBJECT
1 #NAME      (A20)

```

```

1 #STREET   (A20)
1 #CITY     (A20)
1 #SUM      (I4)
1 #MULTI    (I4)
END-DEFINE
*
CREATE OBJECT L-STUDENT OF CLASS 'STUDENTS' /* see METHCL for class
*
L-STUDENT.<> := 'John Smith'
*
SEND METHOD 'INIT' TO L-STUDENT           /* see METHCL
    WITH #VAR1 #VAR2 #VAR3 #VAR4
*
SEND METHOD 'SUMMATION' TO L-STUDENT       /* see METHCL
    WITH #VAR1 #VAR2 #VAR3 #VAR4
*
SEND METHOD 'MULTIPLICATION' TO L-STUDENT /* see METHCL
    WITH #VAR1 #VAR2 #VAR3 #VAR4
*
#NAME      := L-STUDENT.<>
#SUM       := L-STUDENT.<>
#MULTI     := L-STUDENT.<>
*
SEND METHOD 'ADDRESS' TO L-STUDENT        /* see METHCL
*
#STREET    := L-STUDENT.<>
#CITY      := L-STUDENT.<>
*
*
WRITE 'Name   :' #NAME
WRITE 'Street:' #STREET
WRITE 'City   :' #CITY
WRITE ' '
WRITE 'The summation of      ' #VAR1 #VAR2 #VAR3 #VAR4
WRITE 'is' #SUM
WRITE 'The multiplication of' #VAR1 #VAR2 #VAR3 #VAR4
WRITE 'is' #MULTI
*
END

```

Class Definition METHCL Used by METH01:

```

** Example 'METHCL': DEFINE CLASS (used by METH01)
*****
* Defining class STUDENTS for METH01
*
DEFINE CLASS STUDENTS
  OBJECT
    USING METHO           /* Object data for class STUDENTS
  /*
  INTERFACE STUDENT-ARITHMETICS
    PROPERTY FULL-NAME
      IS NAME
    END-PROPERTY
    PROPERTY SUM
    END-PROPERTY
    PROPERTY MULTI
    END-PROPERTY
  *
  METHOD INIT
    IS METH02
    PARAMETER USING METHA

```

```

END-METHOD
METHOD SUMMATION
  IS METH03
  PARAMETER USING METHA
END-METHOD
METHOD MULTIPLICATION
  IS METH04
  PARAMETER USING METHA
END-METHOD
END-INTERFACE
*
INTERFACE STUDENT-ADDRESS
  PROPERTY STUDENT-NAME
    IS NAME
  END-PROPERTY
  PROPERTY STREET
  END-PROPERTY
  PROPERTY CITY
  END-PROPERTY
*
METHOD ADDRESS
  IS METH05
  END-METHOD
END-INTERFACE
END-CLASS
END
    
```

Local Data Area METHO (object data) Used by Class METHCL and Subprograms METH02, METH03, METH04 and METH05:

Local Command	METHO	Library SYSEXSYN	DBID	10 FNR	32
I T L	Name	F	Length	Miscellaneous	> +
All	---	-----	-	-----	----->
	1 NAME	A	20		
	1 STREET	A	30		
	1 CITY	A	20		
	1 SUM	I	4		
	1 MULTI	I	4		

Parameter Data Area METHA Used by Program METH01, Class METHCL and Subprograms METH02, METH03 and METH04:

Parameter Command	METHA	Library SYSEXSYN	DBID	10 FNR	32
I T L	Name	F	Length	Miscellaneous	> +
All	---	-----	-	-----	----->
	1 #VAR1	I	4		
	1 #VAR2	I	4		
	1 #VAR3	I	4		
	1 #VAR4	I	4		

Subprogram METH02 - Method INIT Used by Program METH01:

```

** Example 'METH02': Method INIT (used by METH01)
*****
DEFINE DATA
PARAMETER
  USING METHA
OBJECT
  USING METHO
    
```

```

END-DEFINE
*
* Method INIT of class STUDENTS
*
#VAR1 := 1
#VAR2 := 2
#VAR3 := 3
#VAR4 := 4
*
END

```

Subprogram METH03 - Method SUMMATION Used by Program METH01:

```

** Example 'METH03': Method SUMMATION (used by METH01)
*****
DEFINE DATA
PARAMETER
    USING METHA
OBJECT
    USING METHO
END-DEFINE
*
* Method SUMMATION of class STUDENTS
*
COMPUTE SUM = #VAR1 + #VAR2 + #VAR3 + #VAR4
END

```

Subprogram METH04 - Method MULTIPLICATION Used by Program METH01:

```

** Example 'METH04': Method MULTIPLICATION (used by METH01)
*****
DEFINE DATA
PARAMETER
    USING METHA
OBJECT
    USING METHO
END-DEFINE
*
* Method MULTIPLICATION of class STUDENTS
*
COMPUTE MULTI = #VAR1 * #VAR2 * #VAR3 * #VAR4
END

```

Subprogram METH05 - Method ADDRESS Used by Program METH01:

```

** Example 'METH05': Method ADDRESS (used by METH01)
*****
DEFINE DATA
    OBJECT USING METHO
END-DEFINE
*
* Method ADDRESS of class STUDENTS
*
IF NAME = 'John Smith'
    STREET := 'Oxford street'
    CITY   := 'London'
END-IF
END

```


Output of Program METH01:

Page 1

05-01-17 15:59:04

Name : John Smith
Street: Oxford street
City : London

The summation of	1	2	3	4
is 10				
The multiplication of	1	2	3	4
is 24				